SEMI-CONDUCTOR DEVICES: CHARACTERISTIC PLOTS

Version 1 - June 1995

TABLE

INTRODUCTION

A semi-conductor device is partly described by its voltage-current characteristic. The purpose of this application note is to point out the SMASH^{TM} simulator capabilities to obtain these data interactively. To illustrate our purpose, we present classical semi-conductor devices which are the bipolar and the MOS transistors.

BIPOLAR TRANSISTOR CHARACTERISTICS

First of all we present the usual symbol for a npn transistor, with the Collector, Base and Emitter nodes.



Fig 1: bipolar transistor.

We list below the netlist and pattern files (input SMASH^{TM} files) for the bipolar example. Different simulations are feasible with the following pattern file. We will first see how to obtain the voltage-current characteristics, then we will give examples showing how to access some internal model variables.

```
*---------------------------bip.nsx---
* Bipolar transistor characteristics
*
*-------------------------------------

Q1 Col Base 0 Q2N2222 1
RC VDD Col 30
```

Q1 is an instance of the Q2n2222 model described with the .MODEL syntax.

For each DC analysis,
the command line is written in the pattern file manually or in an interactive way with the SMASH™ Analysis>DC>parameters menu.

Successive analysis
can be asked with the variation of one parameter (declared with .PARAM), thanks to the .PARAMSWEEP directive. The sweep run is then actionable through the SMASH™ analysis menu.

The desired signals may be
gathered into one graph,
or displayed in separated graphs.

```
*------------------------bip.pat---
* Bipolar transistor characteristics
*
*----------------------------------
* SOURCES
VDD VDD 0 10
VC Col 0 Vcol
VB Base 0 Vbase
.PARAM Vbase = 0.8
.PARAM Vcol = 5

.MODEL Q2N2222 NPN IS=14.34F XTI=3
+EG=1.11 VAF=74.03 BF=255.9 NE=1.307
+ISE=14.34F IKF=.2847 XTB=1.5
+BR=6.092 NC=2 ISC=0 IKR=0 RC=1
+CJC=7.306P MJC=.3416 VJC=.75 FC=.5
+CJE=22.01P MJE=.377 VJE=.75
+TR=46.91N TF=411.1P ITF=.6 VTF=1.7
+ XTF=3 RB=10


* SCREEN
* voltage-current characteristics
*---------------------------------
.DC LIN VC 0 10 100m
.PARAMSWEEP Vbase 0.6 0.9 0.05
.TRACE  DC     IC(Q1) I(RC)

*Static and dynamic gain, frequency
*transition are internal model
*parameters.
*---------------------------------
*.DC LIN VB 150m 1.5 10m
*.TRACE  DC  IN(Q1.BETAAC) IN(Q1.BETADC)
*.TRACE  DC     IN(Q1.FT)
```

Voltage-current characteristics curves

In order to visualize the characteristics curves of the transistor, we must observe the current through the collector, vs. the collector voltage, and vary the voltage between the base and the emitter.

The RC resistor will allow us to fix the bias point, by choosing an adequate value. As a matter of fact, for a given collector voltage, the operating point is on the intersection point of curves I(RC) and IC(VCE). The IC vs. V(C) plot is displayed with SMASH™ thanks to a DC analysis on VC, coupled with a sweep analysis on VB. This corresponds to the following lines in the pattern file:

```
.DC LIN VC 0 10 100m
.PARAMSWEEP Vbase 0.6 0.9 0.05
.TRACE  DC     IC(Q1) I(RC)
```



Fig 2: Voltage-current analysis

The operating point is often chosen so that the collector voltage is equal to half the supply voltage, in order to keep a large operational zone. In this example, a resistor RC approximately equal to 30 Ohm will provide an adequate operating point for VC=5V and IC=160mA.

Gain

Other important parameters for a bipolar transistor are its static (dynamic) gain noted BETADC (BETAAC). These parameters depend on collector and base currents, thus on voltage difference between base and emitter nodes. SMASH™ allows you to plot the variation of these internal variables due to the base voltage.

```
.DC LIN VB 150m 1.5 10m
.TRACE  DC  IN(Q1.BETAAC) IN(Q1.BETADC)
```

They are displayed during a DC analysis on VB with a simple adjunction among desired signals (IN(Q1.BETADC) and IN(Q1.BETAAC) where Q1 is the name of the studied transistor).

The simulation shows the difference between static and dynamic gain. However the general shape points out the gain decrease for small currents and strong ones (strong injection). The maximum gain is reached for a base-emitter voltage (VBE) comprised between 0.6V and 0.8V, which is the usual voltage of the base-emitter junction. Indeed, this junction behaves just like a diode.



Fig 3: Static and dynamic gain curves

Please refer to SMASH™ reference manual to explore all the internal variables (Device models chapter, ‘Accessing the internal variables of a bipolar transistor’).

Frequency transition

One may also want to study the transition frequency of the transistor. As for the gain, this internal variable can be accessed easily with the ‘IN(Q1.FT)’ syntax and thus can be easily displayed. The larger frequency band occurs for usual value of VB (between 0.7V and 0.8V) and rapidly decreases for higher value of VB.



Fig 4: Transition frequency

MOS TRANSISTOR CHARACTERISTICS

A similar approach is presented for the MOS transistor.



Fig 5 : MOS transistor

We briefly present SMASH™ input files, describing some possible simulations.

```
*---------------------------mos.nsx---
* Mos transistor caracteristics
*
*-------------------------------------

M1 Drain Gate Source 0 NMOS L=2U W=5U
```

The declaration of this variable as a parameter allows us to sweep. Thus, coupled with any analysis (like DC analysis in our example), one parameter (gate voltage) takes different values as defined with the .PARAMSWEEP directive: VGate varies from 0 to 5V by 0.5V step (see the SMASH™ reference manual). In fact successive simulations are superposed on the screen, clearly showing the influence of this parameter on the analysis.

Note that one paramsweep directive only can be taken into account for an analysis. If you want to keep track of previous paramsweep analysis in your pattern file, you have to comment the implied lines using the comment character '*' at the beginning of the line.

```
*---------------------------mos.pat---
* Mos transistor characteristics
*
*-----------------------------------
* SOURCES
VD Drain 0 5
VG Gate 0 VGate
VS Source 0 0
.PARAM VGate = 1

.model NMOS nmos (level=1
+ vto=.82 KP=45E-6 cgbo=9e-9
+ gamma=.2)

* SCREEN
* Voltage-current characteristics
*----------------------------------
.PARAMSWEEP VGate 0 5 0.5
.DC LIN VD 0 5 0.1
.TRACE  DC     ID(M1)

*Threshold voltage
* VG>=0.9V otherwise no conduction
*----------------------------------
*.DC LIN VG 0 5 0.1
*.TRACE  DC     ID(M1)
*.TRACE  DC     IN(M1.VTH)

*Conductance (GDS) and
*transconductance (GM) are internal
*model parameters.
*----------------------------------
*.DC LIN VG 0.5 4 20m
*.TRACE  DC     IN(M1.GDS)
*.TRACE  DC     IN(M1.GM)
```

Voltage-current characteristics curves

Thanks to a DC analysis on the Drain voltage, the voltage-current characteristics is immediately displayed. The basic simulation shows the drain current versus the drain voltage, and this simulation is successively run for different gate voltage values. The corresponding directives are:

```
.PARAMSWEEP VGate 0 5 0.5
.DC LIN VD 0 5 0.1
.TRACE  DC     ID(M1)
```

The plots below clearly show two operational zones for the mos transistor: linear region and saturation region.



Fig 6: Voltage-current characteristics

Voltage threshold

At first order, the MOS transistor behaves like a switch, driven by the gate voltage. Below a threshold voltage of approximately 1V, there is merely no conduction through the device. This threshold value is pointed out with a DC analysis on the gate voltage, with the visualization of the drain current.

```
.DC LIN VG 0 5 0.1
.TRACE  DC     ID(M1)
```

Moreover, internal model parameters are accessible, just like for bipolar transistor. In particular the threshold voltage can be directly displayed. Thus we can get the exact threshold value (of 820mV), and clearly see that the drain current reaches a significant value for a voltage of 1V at least.

The syntax to visualize these internal parameters is always IN(Mos-instance-name **.** intern-parameter):

```
.DC LIN VG 0.5 4 20m
.TRACE  DC     IN(M1.VTH)
```



Fig 7: Voltage threshold

Internal variables

We just give two examples of access to internal variables, with the conductance and the transconductance of the transistor. The user has to manually enter these TRACE directives in the pattern file. In our example the pattern file should contain these lines:

```
.DC LIN VG 0.5 4 20m
.TRACE  DC     IN(M1.GDS)
.TRACE  DC     IN(M1.GM)
```



Fig 8: Visualization of internal variables

Please report to SMASH^{TM} reference manual (Device models chapter, ‘Accessing the internal variables of a MOS transistor’) to explore the internal variables of the different MOS model.

CONCLUSION

The SMASH^{TM} capabilities allow to study physical characteristics of semi-conductor devices in an interactive way. The direct access to internal variables of device's model is a precious option to rapidly understand influence of inner characteristics on electrical performances.